# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=papilo
pkgver=2.1.2
pkgrel=2
pkgdesc='Parallel Presolve for Integer and Linear Optimization'
arch=(loong64 x86_64)
url='https://www.scipopt.org/'
license=(Apache)
depends=(tbb gmp boost-libs)
makedepends=(cmake boost)
source=(https://github.com/scipopt/$pkgname/archive/v$pkgver/$pkgname-$pkgver.tar.gz)
sha256sums=('7e3d829c957767028db50b5c5085601449b00671e7efc2d5eb0701a6903d102f')

build() {
  cmake -B build -S $pkgname-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=None
  cmake --build build
}

package() {
  DESTDIR="$pkgdir" cmake --install build
}
